Example 7: General Linear Models with Multiple-Slope Interactions of Categorical Predictors in SAS and STATA

This example comes from Hoffman (2015) chapter 2, which examined prediction of cognition (as measured by an information test outcome) from age (centered at 85 years) grip strength (centered at 9 pounds), sex (with men as the reference group) and subsequent dementia status (none = 1, future = 2, and current = 3) in a sample of 550 older adults. Building on the combined final model of Example 6 that included an interaction of age and grip strength, this example shows three coding methods to estimate an interaction between two categorical predictors, sex and dementia group: manual contrasts for both, for one, or none (via program contrasts instead).

SAS Syntax for Importing and Preparing Data for Analysis:

```
* Defining global variable for file location to be replaced in code below;
* \\Client\ precedes path in Virtual Desktop outside H drive;
  %LET filesave= C:\Dropbox\20FA_PSQF6242\PSQF6242_Example7;
* Location for SAS files for these models (uses macro variable filesave);
  LIBNAME filesave "&filesave.";

* Creating value labels to use with categorical predictors;
PROC FORMAT; VALUE FDemGroup 1="1None" 2="2Future" 3="3Current";
             VALUE FSex 0="0Men" 1="1Women"; RUN;

* Import chapter 2 example data into work library as Example7;
DATA work.Example7; SET filesave.SAS_Chapter2;
* Center quantitative predictors;
  age85 = age - 85;
  grip9 = grip - 9;
* Create dummy-coded binary predictors for dementia groups;
  demNF=.; demNC=.; * Create two new empty variables;
  IF demgroup=1 THEN DO; demNF=0; demNC=0; END; * Replace each for none group;
  IF demgroup=2 THEN DO; demNF=1; demNC=0; END; * Replace each for future group;
  IF demgroup=3 THEN DO; demNF=0; demNC=1; END; * Replace each for current group;
* Label new variables - note semi-colon is only at the end of ALL labels;
  LABEL
  age85=     "age85: Age in Years (0=85)"
  grip9=     "grip9: Grip Strength in Pounds (0=9)"
  sexMW=     "sexMW: Sex (0=M, 1=W)"
  demNF=     "demNF: Dementia Contrast for None=0 vs Future=1"
  demNC=     "demNC: Dementia Contrast for None=0 vs Current=1"
  cognition= "cognition: Cognition Outcome"
  demgroup=  "demgroup: Dementia Group 1N 2F 3C";
* Add value labels defined above to categorical predictors;
  FORMAT sexMW Fsex. demgroup Fdemgroup.;
* Select cases complete on variables;
  IF NMISS(cognition,age,grip,sexmw,demgroup)>0 THEN DELETE;
RUN;
```

STATA Syntax for Importing and Preparing Data for Analysis:

```
// Defining global variable for file location to be replaced in code below
// \\Client\ precedes path in Virtual Desktop outside H drive
   global filesave "C:\Dropbox\20FA_PSQF6242\PSQF6242_Example7"

// Import chapter 2 data into temporary file and center predictors
   use "$filesave\STATA_Chapter2.dta", clear // Has converted all variables to lower-case

// Center quantitative predictors
   gen age85 = age - 85
   gen grip9 = grip - 9

// Create dummy-coded binary predictors for dementia groups
   gen demnf=.
   gen demnc=.
// Demgroup = none
   replace demnf=0 if demgroup==1
   replace demnc=0 if demgroup==1
// Demgroup = future
   replace demnf=1 if demgroup==2
   replace demnc=0 if demgroup==2
```

```
// Demgroup = current
   replace demnf=0 if demgroup==3
   replace demnc=1 if demgroup==3

// Add value labels
   label define fdemgroup 1 "1None" 2 "2Future" 3 "3Current"
   label values demgroup fdemgroup
   label define fsex 0 "0Men" 1 "1Women"
   label values sexmw fsex

// Label all variables
   label variable age85     "age85: Age in Years (0=85)"
   label variable grip9     "grip9: Grip Strength in Pounds (0=9)"
   label variable sexmw     "sexmw: Sex (0=Men, 1=Women)"
   label variable demnf     "demnf: Dementia Contrast for None=0 vs Future=1"
   label variable demnc     "demnc: Dementia Contrast for None=0 vs Current=1"
   label variable cognition "cognition: Cognition Outcome"
   label variable demgroup  "demgroup: Dementia Group 1N 2F 3C"
// Select cases complete on variables
   egen nmiss=rowmiss(cognition age grip sexmw demgroup)
   drop if nmiss>0
```

Renouncing ANOVA dogma: You *can* ask for Type I error correction to follow-ups, although I don't believe they should have any special status relative to any other default-provided *p*-values because all group differences and their SEs are already predicted by the model. In addition, it is possible to have significant pairwise group comparisons even if the "omnibus" test is not significant because it is evaluated per DF. So if one group is very different than all the others, then the average effect per DF may not be different than 0 (but you'd be missing the real story by only examining the "omnibus" result). So I do not think the omnibus *F*-test should be the gate-keeper to examining group comparisons. But just remember, for every 20 *p*-values, one is significant by chance!

Equation 2.13, adding sex*dementia group:

$$Cognition_i = \beta_0 + \beta_1(Age_i - 85) + \beta_2(Grip_i - 9) + \beta_3(SexMW_i) + \beta_4(DemNF_i) + \beta_5(DemNC_i) + \beta_6(Age_i - 85)(Grip_i - 9) + \beta_7(SexMW_i)(DemNF_i) + \beta_8(SexMW_i)(DemNC_i) + e_i$$

Dementia Group	Men	Women	Marginal Mean
None	29.07	26.20	27.63
Future	23.01	20.30	21.66
Current	17.10	6.35	11.72
Marginal Mean	23.03	17.62	$\sigma_e^2 = 85.97$

SAS Syntax: Sex and dementia group are both represented via manual dummy-code contrasts

```
TITLE1 "SAS Eq 2.13: Adding Sex by Dementia Interaction";
TITLE2 "Manual Contrasts for Sex (0=Men) and Dementia (0=None)";
PROC GLM DATA=work.Example7 NAMELEN=100;
MODEL cognition = age85 grip9 age85*grip9 sexMW demNF demNC
                  sexMW*demNF sexMW*demNC / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;

CONTRAST "Omnibus DF=2 Dementia*Sex Interaction Test"         sexMW*demNF 1, sexMW*demNC 1;
* In CONTRASTs below, linear combinations are created within a comma set (still 1 DF);
CONTRAST "Omnibus DF=2 Dementia Simple Main Effect for Men"   demNF 1 sexMW*demNF 0,
                                                              demNC 1 sexMW*demNC 0;
CONTRAST "Omnibus DF=2 Dementia Simple Main Effect for Women" demNF 1 sexMW*demNF 1,
                                                              demNC 1 sexMW*demNC 1;
* Request columns of predicted outcome and SE for all cases;
OUTPUT OUT=work.PredOutcomes PREDICTED=Yhat STDP=SEyhat;

* Adjusted cell means (predicted outcomes for age=85 and grip=9);
ESTIMATE "Mean: Men   None"       intercept 1 sexMW 0 demNF  0 demNC 0 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE "Mean: Women None"       intercept 1 sexMW 1 demNF  0 demNC 0 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE "Mean: Men   Future"     intercept 1 sexMW 0 demNF  1 demNC 0 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE "Mean: Women Future"     intercept 1 sexMW 1 demNF  1 demNC 0 sexMW*demNF 1 sexMW*demNC 0;
ESTIMATE "Mean: Men   Current"    intercept 1 sexMW 0 demNF  0 demNC 1 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE "Mean: Women Current"    intercept 1 sexMW 1 demNF  0 demNC 1 sexMW*demNF 0 sexMW*demNC 1;
```

```
* DF=1 simple effects of sex per dementia group;
ESTIMATE "Sex Diff for None"                 sexMW 1 demNF  0 demNC 0 sexMW*demNF  0 sexMW*demNC 0;
ESTIMATE "Sex Diff for Future"               sexMW 1 demNF  0 demNC 0 sexMW*demNF  1 sexMW*demNC 0;
ESTIMATE "Sex Diff for Current"              sexMW 1 demNF  0 demNC 0 sexMW*demNF  0 sexMW*demNC 1;
* DF=1 simple effects dementia group per sex;
ESTIMATE "None-Future Diff for Men"          sexMW 0 demNF  1 demNC 0 sexMW*demNF  0 sexMW*demNC 0;
ESTIMATE "None-Future Diff for Women"        sexMW 0 demNF  1 demNC 0 sexMW*demNF  1 sexMW*demNC 0;
ESTIMATE "None-Current Diff for Men"         sexMW 0 demNF  0 demNC 1 sexMW*demNF  0 sexMW*demNC 0;
ESTIMATE "None-Current Diff for Women"       sexMW 0 demNF  0 demNC 1 sexMW*demNF  0 sexMW*demNC 1;
ESTIMATE "Future-Current Diff for Men"       sexMW 0 demNF -1 demNC 1 sexMW*demNF  0 sexMW*demNC 0;
ESTIMATE "Future-Current Diff for Women"     sexMW 0 demNF -1 demNC 1 sexMW*demNF -1 sexMW*demNC 1;
* DF=1 differences in simple effects = interactions;
ESTIMATE "A: Sex Effect differ between None and Future?"              sexMW*demNF  1 sexMW*demNC 0;
ESTIMATE "A: None-Future Effect differ by Sex?"                       sexMW*demNF  1 sexMW*demNC 0;
ESTIMATE "B: Sex Effect differ between None and Current?"             sexMW*demNF  0 sexMW*demNC 1;
ESTIMATE "B: None-Current Effect differ by Sex?"                      sexMW*demNF  0 sexMW*demNC 1;
ESTIMATE "C: Sex Effect differ between Future and Current?"           sexMW*demNF -1 sexMW*demNC 1;
ESTIMATE "C: Future-Current Effect differ by Sex?"                    sexMW*demNF -1 sexMW*demNC 1;

ODS OUTPUT Estimates=work.SavedEstimates; * Save to compute effect sizes;
RUN; QUIT; TITLE1; TITLE2;
```

STATA Syntax: Sex and dementia group are both represented <u>via manual contrasts</u>

```
display "STATA Eq 2.13: Adding Sex by Dementia Interaction"
display "Manual Contrasts for Sex (0=Men) and Dementia (0=None)"
regress cognition c.age85 c.grip9 c.age85#c.grip9 c.sexmw c.demnf c.demnc ///
                  c.sexmw#c.demnf c.sexmw#c.demnc, level(95)

// Omnibus DF=2 Dementia*Sex Interaction Test
test (c.sexmw#c.demnf=0) (c.sexmw#c.demnc=0)
// In TESTs below, linear combinations are created within parentheses (still 1 DF each)
   // Omnibus DF=2 Dementia Simple Main Effect for Men
   test (c.demnf*1 + c.sexmw#c.demnf*0 =0)(c.demnc*1 + c.sexmw#c.demnc*0=0)
   // Omnibus DF=2 Dementia Simple Main Effect for Women
   test (c.demnf*1 + c.sexmw#c.demnf*1 =0)(c.demnc*1 + c.sexmw#c.demnc*1=0)

// Adjusted cell means (predicted outcomes for age=85 and grip=9)
margins, at(c.age85=0 c.grip9=0 c.sexmw=(0(1)1) c.demnf=0 c.demnc=0) // Means for None
margins, at(c.age85=0 c.grip9=0 c.sexmw=(0(1)1) c.demnf=1 c.demnc=0) // Means for Future
margins, at(c.age85=0 c.grip9=0 c.sexmw=(0(1)1) c.demnf=0 c.demnc=1) // Means for Current

// DF=1 simple effects of sex per dementia group
lincom c.sexmw*1 + c.sexmw#c.demnf*0 + c.sexmw#demnc*0  // Sex Diff for No Dementia
lincom c.sexmw*1 + c.sexmw#c.demnf*1 + c.sexmw#demnc*0  // Sex Diff for Future Dementia
lincom c.sexmw*1 + c.sexmw#c.demnf*0 + c.sexmw#demnc*1  // Sex Diff for Current Dementia
// DF=1 simple effects of dementia group per sex
lincom c.demnf*1  + c.demnc*0 + c.sexmw#c.demnf*0  + c.sexmw#c.demnc*0 // None-Future Diff for Men
lincom c.demnf*1  + c.demnc*0 + c.sexmw#c.demnf*1  + c.sexmw#c.demnc*0 // None-Future Diff for Women
lincom c.demnf*0  + c.demnc*1 + c.sexmw#c.demnf*0  + c.sexmw#c.demnc*0 // None-Current Diff for Men
lincom c.demnf*0  + c.demnc*1 + c.sexmw#c.demnf*0  + c.sexmw#c.demnc*1 // None-Current Diff for Women
lincom c.demnf*-1 + c.demnc*1 + c.sexmw#c.demnf*0  + c.sexmw#c.demnc*0 // Future-Current Diff for Men
lincom c.demnf*-1 + c.demnc*1 + c.sexmw#c.demnf*-1 + c.sexmw#c.demnc*1 // Future-Current Diff for Women

// DF=1 differences in simple effects = interactions
lincom c.sexmw#c.demnf*1  + c.sexmw#c.demnc*0  // A: Sex Effect differ between None and Future?
lincom c.sexmw#c.demnf*1  + c.sexmw#c.demnc*0  // A: None-Future Effect differ by Sex?
lincom c.sexmw#c.demnf*0  + c.sexmw#c.demnc*1  // B: Sex Effect differ between None and Current?
lincom c.sexmw#c.demnf*0  + c.sexmw#c.demnc*1  // B: None-Current Effect differ by Sex?
lincom c.sexmw#c.demnf*-1 + c.sexmw#c.demnc*1  // C: Sex Effect differ between Future and Current?
lincom c.sexmw#c.demnf*-1 + c.sexmw#c.demnc*1  // C: Future-Current Effect differ by Sex?
```

SAS Model Output:

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	**8**	19785.46147	2473.18268	**28.77**	<.0001
Error	**541**	46511.07671	**85.97242**		
Corrected Total	**549**	66296.53818			

R-Square	Coeff Var	Root MSE	cognition Mean
0.298439	37.35476	9.272131	24.82182

Table of Sums of Squares by Model Effect (normally omitted, but shown here to watch how it changes)

Source	DF	Type III SS	Mean Square	F Value	Pr > F	
age85	1	670.468975	670.468975	7.80	0.0054	
grip9	1	1496.911055	1496.911055	17.41	<.0001	
age85*grip9	1	787.786754	787.786754	9.16	0.0026	
sexMW	1	695.197526	695.197526	8.09	0.0046	simple effect for none
demNF	1	1179.272697	1179.272697	13.72	0.0002	simple effect for men
demNC	1	2444.475345	2444.475345	28.43	<.0001	simple effect for women
sexMW*demNF	1	0.541172	0.541172	0.01	0.9368	
sexMW*demNC	1	582.845678	582.845678	6.78	0.0095	

	Total Variation Accounted For				
	Semipartial	Semipartial	Conservative		
Source	Eta-Square	Omega-Square	95% Confidence	Limits	
age85	0.0101	0.0088	0.0002	0.0330	ok but conditional
grip9	0.0226	0.0213	0.0045	0.0526	ok but conditional
age85*grip9	0.0119	0.0106	0.0006	0.0360	ok, unconditional
sexMW	0.0105	0.0092	0.0003	0.0336	ok but conditional
demNF	0.0178	0.0165	0.0025	0.0455	**not ok, needs combined**
demNC	0.0369	0.0355	0.0121	0.0725	**not ok, needs combined**
sexMW*demNF	0.0000	-0.0013	0.0000	0.0027	**not ok, needs combined**
sexMW*demNC	0.0088	0.0075	0.0000	0.0306	**not ok, needs combined**
Omnibus DF=2 Dementia*Sex Interaction Test	0.0091	0.0065	0.0000	0.0287	→ from CONTRAST
Omnibus DF=2 Dementia Simple Main Effect Men	0.0485	0.0458	0.0179	0.0857	→ from CONTRAST
Omnibus DF=2 Dementia Simple Main Effect Women	0.1379	0.1351	0.0872	0.1894	→ from CONTRAST

Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F
Omnibus DF=2 Dementia*Sex Interaction Test	2	600.420410	300.210205	3.49	0.0311
Omnibus DF=2 Dementia Simple Main Effect for Men	2	3213.270817	1606.635408	18.69	<.0001
Omnibus DF=2 Dementia Simple Main Effect for Women	2	9140.125913	4570.062957	53.16	<.0001

Table of Model-Estimated Fixed Effects (normally is last)

Parameter	Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits		
Intercept	29.07014634	0.74849920	38.84	<.0001	27.59982548	30.54046719	**Beta0**
age85	-0.33479877	0.11988755	-2.79	0.0054	-0.57030091	-0.09929663	**Beta1**
grip9	0.61789286	0.14807943	4.17	<.0001	0.32701175	0.90877397	**Beta2**
age85*grip9	0.12215159	0.04035286	3.03	0.0026	0.04288410	0.20141909	**Beta6**
sexMW	-2.87559408	1.01123720	-2.84	0.0046	-4.86202658	-0.88916158	**Beta3**
demNF	-6.05590147	1.63512607	-3.70	0.0002	-9.26787546	-2.84392748	**Beta4**
demNC	-11.97073055	2.24495370	-5.33	<.0001	-16.38062470	-7.56083640	**Beta5**
sexMW*demNF	0.16426999	2.07047524	0.08	0.9368	-3.90288588	4.23142586	**Beta7**
sexMW*demNC	-7.87509954	3.02453647	-2.60	0.0095	-13.81637383	-1.93382526	**Beta8**

Interpret these fixed effects:

Simple main effect of Sex β_3 =

Simple main effect of Dem None vs Future β_4 =

Simple main effect of Dem None vs Current β_5 =

Interpret Sex by DemNF β_7 → Sex as Simple Effect, DemNF as Moderator:

Interpret Sex by DemNC β_8 → Sex as Simple Effect, DemNC as Moderator:

Interpret Sex by DemNF β_7 → DemNF as Simple Effect, Sex as Moderator:

Interpret Sex by DemNC β_8 → DemNC as Simple Effect, Sex as Moderator:

Parameter		Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits	
Adjusted cell means (for age=85 and grip=9)							
Cognition for Men	None	29.0701463	0.74849920	38.84	<.0001	27.5998255	30.5404672
Cognition for Women	None	26.1945523	0.63883397	41.00	<.0001	24.9396532	27.4494513
Cognition for Men	Future	23.0142449	1.49276013	15.42	<.0001	20.0819286	25.9465611
Cognition for Women	Future	20.3029208	1.11863329	18.15	<.0001	18.1055238	22.5003177
Cognition for Men	Current	17.0994158	2.14021810	7.99	<.0001	12.8952599	21.3035717
Cognition for Women	Current	6.3487222	1.94788049	3.26	0.0012	2.5223863	10.1750580
DF=1 simple effects of sex per dementia group							
Sex Difference for No Dementia		-2.8755941	1.01123720	-2.84	0.0046	-4.8620266	-0.8891616
Sex Difference for Future Dementia		-2.7113241	1.87406883	-1.45	0.1485	-6.3926673	0.9700192
Sex Difference for Current Dementia		-10.7506936	2.89932314	-3.71	0.0002	-16.4460040	-5.0553832
DF=1 simple effects of dementia group per sex							
None-Future Difference for Men		-6.0559015	1.63512607	-3.70	0.0002	-9.2678755	-2.8439275
None-Future Difference for Women		-5.8916315	1.27776082	-4.61	<.0001	-8.4016120	-3.3816510
None-Current Difference for Men		-11.9707305	2.24495370	-5.33	<.0001	-16.3806247	-7.5608364
None-Current Difference for Women		-19.8458301	2.02858306	-9.78	<.0001	-23.8306947	-15.8609655
Future-Current Difference for Men		-5.9148291	2.58676242	-2.29	0.0226	-10.9961581	-0.8335000
Future-Current Difference for Women		-13.9541986	2.23891711	-6.23	<.0001	-18.3522347	-9.5561625
DF=1 differences in simple effects = interactions							
A: Sex Effect differ between None and Fut?		0.1642700	2.07047524	0.08	0.9368	-3.9028859	4.2314259
A: None-Future Effect differ by Sex?		0.1642700	2.07047524	0.08	0.9368	-3.9028859	4.2314259
B: Sex Effect differ between None and Cur?		-7.8750995	3.02453647	-2.60	0.0095	-13.8163738	-1.9338253
B: None-Current Effect differ by Sex?		-7.8750995	3.02453647	-2.60	0.0095	-13.8163738	-1.9338253
C: Sex Effect differ between Fut and Cur?		-8.0393695	3.41516478	-2.35	0.0189	-14.7479779	-1.3307612
C: Future-Current Effect differ by Sex?		-8.0393695	3.41516478	-2.35	0.0189	-14.7479779	-1.3307612

Effects in Gray Italics are not yet relevant (i.e., not given by this model)…

Dementia Group Means	Men Mean	Women Mean	*MARGINAL MEAN*	Sex Difference
None Mean	29.07	26.20	*27.63*	−2.87 (*p*=.0046)
Future Mean	23.01	20.30	*21.66*	−2.71 (*p*=.1485)
Current Mean	17.10	6.35	*11.72*	−10.75 (*p*=.0002)
MARGINAL	*23.03*	*17.62*		*−5.45 (p<.0001)*

Dementia Group Differences	Men Diff	Women Diff	*MARGINAL DIFF*	Simple Effect Difference
None-Future Diff	−6.06 (*p*=.0002)	−5.90 (*p*<.0001)	*−5.97 (p<.0001)*	A =0.16 (*p*=.9368)
None-Current Diff	−11.97 (*p*<.0001)	−19.85 (*p*<.0001)	*−15.91 (p<.0001)*	B= −7.88 (*p*=.0095)
Future-Current Diff	−5.91 (*p*=.0226)	−13.95 (*p*<.0001)	*−9.93 (p<.0001)*	C= −8.04 (*p*=.0189)

```
* Calculate effect sizes from t-values;
DATA work.SavedEstimates; SET work.SavedEstimates; %LET df=541;
     CohenD=(2*tvalue)/SQRT(&df.);
     EffectR=tvalue/SQRT((tvalue*tvalue)+&df.);
RUN;
* Print effect sizes;
TITLE "Effect Sizes";
PROC PRINT NOOBS DATA=work.SavedEstimates;
     WHERE INDEX(Parameter,"Cognition")=0; * Remove intercepts;
     VAR Parameter Estimate StdErr probt CohenD EffectR; RUN; TITLE;
```

Effect Sizes

Parameter	Estimate	StdErr	Probt	CohenD	EffectR
Sex Difference for No Dementia	-2.8755941	1.01123720	0.0046	-0.24452	-0.12135
Sex Difference for Future Dementia	-2.7113241	1.87406883	0.1485	-0.12440	-0.06208
Sex Difference for Current Dementia	-10.7506936	2.89932314	0.0002	-0.31884	-0.15743
None-Future Difference for Men	-6.0559015	1.63512607	0.0002	-0.31846	-0.15725
None-Future Difference for Women	-5.8916315	1.27776082	<.0001	-0.39648	-0.19445
None-Current Difference for Men	-11.9707305	2.24495370	<.0001	-0.45851	-0.22346
None-Current Difference for Women	-19.8458301	2.02858306	<.0001	-0.84122	-0.38771

Future-Current Difference for Men	-5.9148291	2.58676242	0.0226	-0.19662	-0.09784
Future-Current Difference for Women	-13.9541986	2.23891711	<.0001	-0.53592	-0.25883
A: Sex Effect differ between None and Fut?	0.1642700	2.07047524	0.9368	0.00682	0.00341
A: None-Future Effect differ by Sex?	0.1642700	2.07047524	0.9368	0.00682	0.00341
B: Sex Effect differ between None and Cur?	-7.8750995	3.02453647	0.0095	-0.22389	-0.11125
B: None-Current Effect differ by Sex?	-7.8750995	3.02453647	0.0095	-0.22389	-0.11125
C: Sex Effect differ between Fut and Cur?	-8.0393695	3.41516478	0.0189	-0.20241	-0.10069
C: Future-Current Effect differ by Sex?	-8.0393695	3.41516478	0.0189	-0.20241	-0.10069

```
* Plot saved predicted values for fake people -- dementia as X;
PROC SGPLOT DATA=work.PredOutcomes;
     WHERE PersonID=-99; * Only for fake people;
     SERIES x=demgroup y=Yhat / GROUP=sexMW;
     XAXIS GRID LABEL="Dementia Group" VALUES=(1 TO 3 BY 1);
     YAXIS GRID LABEL="Predicted Cognition" VALUES=(0 TO 35 BY 5);
RUN; QUIT;
```




```
* Plot saved predicted values for fake people -- sex as X;
PROC SGPLOT DATA=work.PredOutcomes;
     WHERE PersonID=-99; * Only for fake people;
     SERIES x=sexMW y=Yhat / GROUP=demgroup;
     XAXIS GRID LABEL="" VALUES=(3 TO 15 BY 3);
     YAXIS GRID LABEL="Predicted Cognition" VALUES=(15 TO 45 BY 5);
RUN; QUIT;
```



Same model, but dementia group is "categorical" (represented by program-created contrasts) which changes the dementia reference group to Current

$$Cognition_i = \beta_0 + \beta_1(Age_i - 85) + \beta_2(Grip_i - 9) + \beta_3(SexMW_i) + \beta_4(DemCN_i) + \beta_5(DemCF_i) + \beta_6(Age_i - 85)(Grip_i - 9) + \beta_7(SexMW_i)(DemCN_i) + \beta_8(SexMW_i)(DemCF_i) + e_i$$

SAS Model Syntax:

```
TITLE1 "SAS Eq 2.13: Adding Sex by Dementia Interaction";
TITLE2 "Manual Contrast for Sex (0=Men) and Program-Categorical Dementia (Ref=Current)";
PROC GLM DATA=work.Example7 NAMELEN=100;
CLASS demgroup; * CLASS: demgroup is categorical (program makes dummy codes);
MODEL cognition = age85 grip9 age85*grip9 sexMW demgroup
                  sexMW*demgroup / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;

* First two CONTRASTs below now given by default with use of CLASS;
CONTRAST "Omnibus DF=2 Dementia*Sex Interaction Test"  sexMW*demgroup -1 1 0, sexMW*demgroup -1 0 1;
* In CONTRASTs below, linear combinations are created within a comma set (still 1 DF each);
CONTRAST "Omnibus DF=2 Dementia Simple Main Effect for Men"   demgroup -1 1 0 sexMW*demgroup  0 0 0,
                                                              demgroup -1 0 1 sexMW*demgroup  0 0 0;
CONTRAST "Omnibus DF=2 Dementia Simple Main Effect for Women" demgroup -1 1 0 sexMW*demgroup -1 1 0,
                                                              demgroup -1 0 1 sexMW*demgroup -1 0 1;
* Request columns of predicted outcome and SE for all cases;
OUTPUT OUT=work.PredOutcomes PREDICTED=Yhat STDP=SEyhat;
* Adjusted cell means (at age=85 and grip=9) and simple effects of dementia group diffs per sex;
LSMEANS demgroup / STDERR TDIFF PDIFF ADJUST=T DIFF=ALL AT(age85 grip9 sexMW) = (0 0 0); * For men;
LSMEANS demgroup / STDERR TDIFF PDIFF ADJUST=T DIFF=ALL AT(age85 grip9 sexMW) = (0 0 1); * For women;
* DF=1 simple effects of sex per dementia group (not redundant with LSMEANS);
ESTIMATE "Sex Diff for None"             demgroup  0  0  0  sexMW 1  sexMW*demgroup  1  0  0;
ESTIMATE "Sex Diff for Future"           demgroup  0  0  0  sexMW 1  sexMW*demgroup  0  1  0;
ESTIMATE "Sex Diff for Current"          demgroup  0  0  0  sexMW 1  sexMW*demgroup  0  0  1;
* DF=1 simple effects of dementia group per sex (redundant with LSMEANS);
ESTIMATE "None-Future Diff for Men"      demgroup -1  1  0  sexMW 0  sexMW*demgroup  0  0  0;
ESTIMATE "None-Future Diff for Women"    demgroup -1  1  0  sexMW 0  sexMW*demgroup -1  1  0;
ESTIMATE "None-Current Diff for Men"     demgroup -1  0  1  sexMW 0  sexMW*demgroup  0  0  0;
ESTIMATE "None-Current Diff for Women"   demgroup -1  0  1  sexMW 0  sexMW*demgroup -1  0  1;
ESTIMATE "Future-Current Diff for Men"   demgroup  0 -1  1  sexMW 0  sexMW*demgroup  0  0  0;
ESTIMATE "Future-Current Diff for Women" demgroup  0 -1  1  sexMW 0  sexMW*demgroup  0 -1  1;
* DF=1 differences in simple effects = interactions (not redundant with LSMEANS);
ESTIMATE "A: Sex Effect differ between None and Future?"            sexMW*demgroup -1  1  0;
ESTIMATE "A: None-Future Effect differ by Sex?"                     sexMW*demgroup -1  1  0;
ESTIMATE "B: Sex Effect differ between None and Current?"           sexMW*demgroup -1  0  1;
ESTIMATE "B: None-Current Effect differ by Sex?"                    sexMW*demgroup -1  0  1;
ESTIMATE "C: Sex Effect differ between Future and Current?"         sexMW*demgroup  0 -1  1;
ESTIMATE "C: Future-Current Effect differ by Sex?"                  sexMW*demgroup  0 -1  1;
RUN; QUIT; TITLE1; TITLE2;
```

STATA Model Syntax:

```
display "STATA Eq 2.13: Adding Sex by Dementia Interaction"
display "Manual Contrast for Sex (0=Men) and Program-Categorical Dementia (Ref=Current)"
regress cognition c.age85 c.grip9 c.age85#c.grip9 c.sexmw ib(last).demgroup  ///
                  c.sexmw#ib(last).demgroup, level(95)
// Omnibus DF=2 Dementia*Sex Interaction Test
contrast c.sexmw#i.demgroup
// In TESTs below, linear combinations are created within parentheses (still 1 DF each)
   // Omnibus DF=2 Dementia Simple Main Effect for Men
   test (i1.demgroup*1 + c.sexmw#i1.demgroup*0 =0) (i2.demgroup*1 + c.sexmw#i2.demgroup*0 =0)
   // Omnibus DF=2 Dementia Simple Main Effect for Women
   test (i1.demgroup*1 + c.sexmw#i1.demgroup*1 =0) (i2.demgroup*1 + c.sexmw#i2.demgroup*1 =0)

// Adjusted cell means (predicted outcomes for age=85 and grip=9)
margins  i.demgroup, at(c.sexmw=(0(1)1) c.age85=0 c.grip9=0) vsquish
marginsplot, xdimension(demgroup) // Plot pred outcomes by dementia group
marginsplot, xdimension(sexmw)    // Plot pred outcomes by sex
// Dem pairwise (means, then diffs) for Men
margins  i.demgroup, at(c.sexmw=0 c.age85=0 c.grip9=0) vsquish
margins  i.demgroup, at(c.sexmw=0 c.age85=0 c.grip9=0) vsquish pwcompare(pveffects)
// Dem pairwise (means, then diffs) for Women
margins  i.demgroup, at(c.sexmw=1 c.age85=0 c.grip9=0) vsquish
margins  i.demgroup, at(c.sexmw=1 c.age85=0 c.grip9=0) vsquish pwcompare(pveffects)
```

```
// DF=1 simple effects of sex per dementia group (not redundant with margins)
lincom c.sexmw*1 + c.sexmw#i1.demgroup*1  // Sex diff for None
lincom c.sexmw*1 + c.sexmw#i2.demgroup*1  // Sex diff for Future
lincom c.sexmw*1 + c.sexmw#i3.demgroup*1  // Sex diff for Current
// DF=1 dementia simple group diffs (redundant with margins)
// None-Future Diff for Men
lincom i1.demgroup*-1 +i2.demgroup*1 + c.sexmw#i1.demgroup*0  + c.sexmw#i2.demgroup*0
// None-Future Diff for Women
lincom i1.demgroup*-1 +i2.demgroup*1 + c.sexmw#i1.demgroup*-1 + c.sexmw#i2.demgroup*1
// None-Current Diff for Men
lincom i1.demgroup*-1 +i3.demgroup*1 + c.sexmw#i1.demgroup*0  + c.sexmw#i3.demgroup*0
// None-Current Diff for Women
lincom i1.demgroup*-1 +i3.demgroup*1 + c.sexmw#i1.demgroup*-1 + c.sexmw#i3.demgroup*1
// Fut-Cur Diff for Men
lincom i2.demgroup*-1 +i3.demgroup*1 + c.sexmw#i2.demgroup*0  + c.sexmw#i3.demgroup*0
// Fut-Cur Diff for Women
lincom i2.demgroup*-1 +i3.demgroup*1 + c.sexmw#i2.demgroup*-1 + c.sexmw#i3.demgroup*1
// DF=1 differences in simple effects = interactions (not redundant with margins)
lincom c.sexmw#i1.demgroup*-1 + c.sexmw#i2.demgroup*1 // A: Sex Effect differ btw None and Future?
lincom c.sexmw#i1.demgroup*-1 + c.sexmw#i2.demgroup*1 // A: None-Future Effect differ by Sex?
lincom c.sexmw#i1.demgroup*-1 + c.sexmw#i3.demgroup*1 // B: Sex Effect differ btw None and Current?
lincom c.sexmw#i1.demgroup*-1 + c.sexmw#i3.demgroup*1 // B: None-Current Effect differ by Sex?
lincom c.sexmw#i2.demgroup*-1 + c.sexmw#i3.demgroup*1 // C: Sex Effect differ btw Future and Current?
lincom c.sexmw#i2.demgroup*-1 + c.sexmw#i3.demgroup*1 // C: Future-Current Effect differ by Sex?
```

SAS Model Output (model test and R^2 are same so are omitted):

Table of Sums of Squares by Model Effect (normally omitted, but shown here to watch how it changes)

Source	DF	Type III SS	Mean Square	F Value	Pr > F
age85	1	670.468975	670.468975	7.80	0.0054
grip9	1	1496.911055	1496.911055	17.41	<.0001
age85*grip9	1	787.786754	787.786754	9.16	0.0026
sexMW	**1**	**1672.026897**	**1672.026897**	**19.45**	**<.0001 → now marginal**
demgroup	2	3213.270817	1606.635408	18.69	<.0001 → CLASS automatic
sexMW*demgroup	2	600.420410	300.210205	3.49	0.0311 → CLASS automatic

	Total Variation Accounted For			
	Semipartial	Semipartial Omega-Square	95% Confidence Limits	
Source	Eta-Square			
age85	0.0101	0.0088	0.0002	0.0330
grip9	0.0226	0.0213	0.0045	0.0526
age85*grip9	0.0119	0.0106	0.0006	0.0360
sexMW	**0.0252**	**0.0239**	**0.0058**	**0.0565 → now marginal over demgroup**
demgroup	0.0485	0.0458	0.0179	0.0857 → CLASS automatic (and from CONTRAST)
sexMW*demgroup	0.0091	0.0065	0.0000	0.0287 → CLASS automatic (and from CONTRAST)
demgroup for women	0.1379	0.1351	0.0872	0.1894 → from CONTRAST

Bold: Because demgroup is on CLASS, the *F*-test and sr^2 effect size for the main effect of sex is **now marginal** across dementia, but the *F*-test and sr^2 effect size for the main effect of dementia is still the simple effect for men.

Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F
Omnibus DF=2 Dementia*Sex Interaction Test	2	600.420410	300.210205	3.49	0.0311
Omnibus DF=2 Dementia Simple Main Effect for Men	2	3213.270817	1606.635408	18.69	<.0001
Omnibus DF=2 Dementia Simple Main Effect for Women	2	9140.125913	4570.062957	53.16	<.0001

Table of Model-Estimated Fixed Effects (normally is last)

Parameter		Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits		
Intercept		17.09941579 B	2.14021810	7.99	<.0001	12.89525992	21.30357165	**new Beta0**
age85		-0.33479877	0.11988755	-2.79	0.0054	-0.57030091	-0.09929663	**old Beta1**
grip9		0.61789286	0.14807943	4.17	<.0001	0.32701175	0.90877397	**old Beta2**
age85*grip9		0.12215159	0.04035286	3.03	0.0026	0.04288410	0.20141909	**old Beta6**
sexMW		-10.75069362 B	2.89932314	-3.71	0.0002	-16.44600401	-5.05538323	**new Beta3**
demgroup	1None	11.97073055 B	2.24495370	5.33	<.0001	7.56083640	16.38062470	**new Beta4**
demgroup	2Future	5.91482908 B	2.58676242	2.29	0.0226	0.83350002	10.99615814	**new Beta5**
demgroup	3Current	0.00000000 B	.	.	.	.	.	

```
sexMW*demgroup 1None       7.87509954 B   3.02453647     2.60    0.0095   1.93382526  13.81637383  new Beta7
sexMW*demgroup 2Future     8.03936953 B   3.41516478     2.35    0.0189   1.33076117  14.74797790  new Beta8
sexMW*demgroup 3Current    0.00000000 B    .              .       .        .           .

NOTE: The X'X matrix has been found to be singular, and a generalized inverse was used to solve the normal
equations.  Terms whose estimates are followed by the letter 'B' are not uniquely estimable.
```

Translation: "Hey, I noticed that you only needed two contrasts to distinguish your three demgroup categories, so I didn't put in the third one that would have been redundant. So your reference group is the category that has the dots instead, k? Oh, and the "B" is there to remind you that these are each conditional given the interaction, k?"

```
Least Squares Means at age85=0, grip9=0, sexMW=0
               cognition        Standard                   LSMEAN
demgroup          LSMEAN           Error    Pr > |t|       Number
1None         29.0701463       0.7484992      <.0001            1
2Future       23.0142449       1.4927601      <.0001            2
3Current      17.0994158       2.1402181      <.0001            3

   Least Squares Means for Effect demgroup
   t for H0: LSMean(i)=LSMean(j) / Pr > |t|
i/j              1             2             3
   1                     3.70363      5.332284
                          0.0002        <.0001
   2       -3.70363                   2.286576
             0.0002                     0.0226
   3       -5.33228      -2.28658
             <.0001        0.0226
```

These are the results from the LSMEANS asking for demgroup means and differences for each level of sexMW (0, 1). Note that only t-values and p-values are given (not the estimates of the difference or those SEs).

```
Least Squares Means at age85=0, grip9=0, sexMW=1
               cognition        Standard                   LSMEAN
demgroup          LSMEAN           Error    Pr > |t|       Number
1None         26.1945523       0.6388340      <.0001            1
2Future       20.3029208       1.1186333      <.0001            2
3Current       6.3487222       1.9478805      0.0012            3

   Least Squares Means for Effect demgroup
   t for H0: LSMean(i)=LSMean(j) / Pr > |t|
i/j              1             2             3
   1                    4.610903       9.7831
                          <.0001        <.0001
   2       -4.6109                    6.232566
             <.0001                     <.0001
   3       -9.7831       -6.23257
             <.0001        <.0001
```

Parameter	Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits	
Cell means (simple means) are given via LSMEANS above						
DF=1 simple effects of sex (not given by LSMEANS above)						
Sex Difference for None	-2.8755941	1.01123720	-2.84	0.0046	-4.8620266	-0.8891616
Sex Difference for Future	-2.7113241	1.87406883	-1.45	0.1485	-6.3926673	0.9700192
Sex Difference for Current	-10.7506936	2.89932314	-3.71	0.0002	-16.4460040	-5.0553832
DF=1 simple effects of dementia group (already given by LSMEANS above)						
None-Future Difference for Men	-6.0559015	1.63512607	-3.70	0.0002	-9.2678755	-2.8439275
None-Future Difference for Women	-5.8916315	1.27776082	-4.61	<.0001	-8.4016120	-3.3816510
None-Current Difference for Men	-11.9707305	2.24495370	-5.33	<.0001	-16.3806247	-7.5608364
None-Current Difference for Women	-19.8458301	2.02858306	-9.78	<.0001	-23.8306947	-15.8609655
Future-Current Difference for Men	-5.9148291	2.58676242	-2.29	0.0226	-10.9961581	-0.8335000
Future-Current Difference for Women	-13.9541986	2.23891711	-6.23	<.0001	-18.3522347	-9.5561625
DF=1 differences in simple effects = interactions not given above						
A: Sex Effect differ between None and Fut?	0.1642700	2.07047524	0.08	0.9368	-3.9028859	4.2314259
A: None-Future Effect differ by Sex?	0.1642700	2.07047524	0.08	0.9368	-3.9028859	4.2314259
B: Sex Effect differ between None and Cur?	-7.8750995	3.02453647	-2.60	0.0095	-13.8163738	-1.9338253

```
B: None-Current Effect differ by Sex?        -7.8750995  3.02453647    -2.60    0.0095  -13.8163738   -1.9338253
C: Sex Effect differ between Fut and Cur?    -8.0393695  3.41516478    -2.35    0.0189  -14.7479779   -1.3307612
C: Future-Current Effect differ by Sex?      -8.0393695  3.41516478    -2.35    0.0189  -14.7479779   -1.3307612
```

Same model, but now both sex and dementia group are "categorical" (represented by program-created contrasts with last as reference group in each):

$$Cognition_i = \beta_0 + \beta_1(Age_i - 85) + \beta_2(Grip_i - 9) + \beta_3(SexWM_i) + \beta_4(DemCN_i) + \beta_5(DemCF_i) + \beta_6(Age_i - 85)(Grip_i - 9) + \beta_7(SexWM_i)(DemCN_i) + \beta_8(SexWM_i)(DemCF_i) + e_i$$

SAS Model Syntax:

```
TITLE1 "SAS Eq 2.13: Adding Sex by Dementia Interaction";
TITLE2 "Program-Categorical Sex (Ref=Women) and Program-Categorical Dementia (Ref=Current)";
PROC GLM DATA=work.Example7 NAMELEN=100;
CLASS sexMW demgroup; * CLASS: sexMW and demgroup are categorical;
MODEL cognition = age85 grip9 age85*grip9 sexMW demgroup
                  sexMW*demgroup / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
* CONTRAST for interaction given by default with use of CLASS;
CONTRAST "Omnibus DF=2 Dementia*Sex Interaction Test" sexMW*demgroup -1 1 0 1 -1 0,
                                                      sexMW*demgroup -1 0 1 1 0 -1;
* Request columns of predicted outcome and SE for all cases;
OUTPUT OUT=work.PredOutcomes PREDICTED=Yhat STDP=SEyhat;
* Marginal sex means and tests diffs averaged over demgroup;
LSMEANS sexMW    / STDERR TDIFF PDIFF ADJUST=T DIFF=ALL AT(age85 grip9) = (0 0);
* Marginal demgroup means and tests of diffs averaged over sex (F-test below);
LSMEANS demgroup / STDERR TDIFF PDIFF ADJUST=T DIFF=ALL AT(age85 grip9) = (0 0);
* Traditional ANOVA output ends here;
* Adjusted cell means (at age=85 and grip=9) and omnibus simple effect tests;
LSMEANS sexMW*demgroup / STDERR TDIFF PDIFF ADJUST=T SLICE=demgroup SLICE=sexMW
                         DIFF=ALL AT(age85 grip9) = (0 0);
* Order by CLASS statement: MN MF MC WN WF WC -- check order with / E;
* DF=1 simple effects of sex per dementia group (redundant with LSMEANS);
ESTIMATE 'Sex Diff for None'          demgroup  0  0  0 sexMW -1 1 sexMW*demgroup -1  0  0  1  0  0 /E;
ESTIMATE 'Sex Diff for Fut'           demgroup  0  0  0 sexMW -1 1 sexMW*demgroup  0 -1  0  0  1  0;
ESTIMATE 'Sex Diff for Cur'           demgroup  0  0  0 sexMW -1 1 sexMW*demgroup  0  0 -1  0  0  1;
* DF=1 simple effects of dementia group per sex (redundant with LSMEANS);
ESTIMATE 'None-Fut Diff for Men'      demgroup -1  1  0 sexMW  0 0 sexMW*demgroup -1  1  0  0  0  0;
ESTIMATE 'None-Fut Diff for Women'    demgroup -1  1  0 sexMW  0 0 sexMW*demgroup  0  0  0 -1  1  0;
ESTIMATE 'None-Cur Diff for Men'      demgroup -1  0  1 sexMW  0 0 sexMW*demgroup -1  0  1  0  0  0;
ESTIMATE 'None-Cur Diff for Women'    demgroup -1  0  1 sexMW  0 0 sexMW*demgroup  0  0  0 -1  0  1;
ESTIMATE 'Fut-Cur Diff for Men'       demgroup  0 -1  1 sexMW  0 0 sexMW*demgroup  0 -1  1  0  0  0;
ESTIMATE 'Fut-Cur Diff for Women'     demgroup  0 -1  1 sexMW  0 0 sexMW*demgroup  0  0  0  0 -1  1;
* DF=1 differences in simple effects = interactions (not redundant with LSMEANS);
ESTIMATE "A: Sex Effect differ between None and Future?"          sexMW*demgroup -1  1  0  1 -1  0;
ESTIMATE "A: None-Future Effect differ by Sex?"                   sexMW*demgroup -1  1  0  1 -1  0;
ESTIMATE "B: Sex Effect differ between None and Current?"         sexMW*demgroup -1  0  1  1  0 -1;
ESTIMATE "B: None-Cur Effect differ by Sex?"                      sexMW*demgroup -1  0  1  1  0 -1;
ESTIMATE "C: Sex Effect differ between Future and Current?"       sexMW*demgroup  0 -1  1  0  1 -1;
ESTIMATE "C: Future-Current Effect differ by Sex?"                sexMW*demgroup  0 -1  1  0  1 -1;
RUN; QUIT; TITLE1; TITLE2;
```

STATA Model Syntax:

```
display "STATA Eq 2.13: Adding Sex by Dementia Interaction"
display "Program-Categorical Sex (Ref=Women) and Program-Categorical Dementia (Ref=Current)"
regress cognition c.age85 c.grip9 c.age85#c.grip9 ib(last).sexmw ib(last).demgroup ///
                  ib(last).sexmw#ib(last).demgroup, level(95)
contrast i.sexmw            // Omnibus DF=1 Marginal Sex Main Effect (not given by default)
contrast i.demgroup         // Omnibus DF=2 Marginal Dementia Main Effect (not given by default)
contrast i.sexmw#i.demgroup // Omnibus Df=2 Dementia*Sex Interaction Test (not given by default)
// Marginal means (then diffs) for Sex
margins  i.sexmw, asbalanced at(c.age85=0 c.grip9=0) vsquish
margins  i.sexmw, asbalanced at(c.age85=0 c.grip9=0) vsquish pwcompare(pveffects)
// Marginal means (then diffs) for Dementia
margins  i.demgroup, asbalanced at(c.age85=0 c.grip9=0) vsquish
margins  i.demgroup, asbalanced at(c.age85=0 c.grip9=0) vsquish pwcompare(pveffects)
// Traditional ANOVA output ends here
// Adjusted cell means (at age=85 and grip=9) then diffs
margins  i.sexmw#i.demgroup, at(c.age85=0 c.grip9=0) vsquish
margins  i.sexmw#i.demgroup, at(c.age85=0 c.grip9=0) vsquish pwcompare(pveffects)
```

```
margins  i.sexmw@i.demgroup, at(c.age85=0 c.grip9=0) vsquish
marginsplot, xdimension(demgroup) // Plot pred outcomes by dementia group
marginsplot, xdimension(sexmw)    // Plot pred outcomes by sex
// DF=1 simple effect of sex per demgroup (not redundant with margins)
margins  i.sexmw@i.demgroup, at(c.age85=0 c.grip9=0)
// DF=2 simple effect of demgroup per sex (not redundant with margins)
margins  i.demgroup@i.sexmw, at(c.age85=0 c.grip9=0)
// DF=1 differences in simple effects = interactions (not redundant with margins)
contrast {i.sexmw#i.demgroup  1 -1  0 -1  1  0} // A: Sex Effect differ between None and Future?
contrast {i.sexmw#i.demgroup  1 -1  0 -1  1  0} // A: None-Future Effect differ by Sex?
contrast {i.sexmw#i.demgroup  1  0 -1 -1  0  1} // B: Sex Effect differ between None and Current?
contrast {i.sexmw#i.demgroup  1  0 -1 -1  0  1} // B: None-Current Effect differ by Sex?
contrast {i.sexmw#i.demgroup  0  1 -1  0 -1  1} // C: Sex Effect differ between Future and Current?
contrast {i.sexmw#i.demgroup  0  1 -1  0 -1  1} // C: Future-Current Effect differ by Sex?
```

SAS Model Output (model test and R^2 are same so are omitted):

Table of Sums of Squares by Model Effect (normally omitted, but shown here to watch how it changes)

Source	DF	Type III SS	Mean Square	F Value	Pr > F
age85	1	670.46897	670.46897	7.80	0.0054
grip9	1	1496.91105	1496.91105	17.41	<.0001
age85*grip9	1	787.78675	787.78675	9.16	0.0026
sexMW	1	1672.02690	1672.02690	19.45	<.0001
demgroup	**2**	**11111.46301**	**5555.73151**	**64.62**	**<.0001**
sexMW*demgroup	2	600.42041	300.21020	3.49	0.0311

	Total Variation Accounted For			
	Semipartial	Semipartial Omega-Square	95% Confidence Limits	
Source	Eta-Square			
age85	0.0101	0.0088	0.0002	0.0330
grip9	0.0226	0.0213	0.0045	0.0526
age85*grip9	0.0119	0.0106	0.0006	0.0360
sexMW	0.0252	0.0239	0.0058	0.0565
demgroup	**0.1676**	**0.1648**	**0.1132**	**0.2213**
sexMW*demgroup	0.0091	0.0065	0.0000	0.0287

Bold: Because demgroup is on CLASS, the *F*-test and sr^2 effect size for the main effect of sex is **marginal** across dementia. Because sexMW is also now on CLASS, the *F*-test and sr^2 effect size for the main effect of dementia is now marginal across genders.

These are the sr^2 values in the results.

Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F
Omnibus DF=1 Sex Marginal Main Effect	1	1672.02690	1672.02690	19.45	<.0001

Table of Model-Estimated Fixed Effects (normally is last)

Parameter		Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits		
Intercept		6.34872217 B	1.94788049	3.26	0.0012	2.52238633	10.17505800	**new Beta0**
age85		-0.33479877	0.11988755	-2.79	0.0054	-0.57030091	-0.09929663	**old Beta1**
grip9		0.61789286	0.14807943	4.17	<.0001	0.32701175	0.90877397	**old Beta2**
age85*grip9		0.12215159	0.04035286	3.03	0.0026	0.04288410	0.20141909	**old Beta6**
sexMW	OMen	10.75069362 B	2.89932314	3.71	0.0002	5.05538323	16.44600401	**new Beta3**
sexMW	1Women	0.00000000 B	.	.	.	.	.	
demgroup	1None	19.84583009 B	2.02858306	9.78	<.0001	15.86096548	23.83069470	**new Beta4**
demgroup	2Future	13.95419861 B	2.23891711	6.23	<.0001	9.55616249	18.35223473	**new Beta5**
demgroup	3Current	0.00000000 B	.	.	.	.	.	
sexMW*demgroup	OMen 1None	-7.87509954 B	3.02453647	-2.60	0.0095	-13.81637383	-1.93382526	**new Beta7**
sexMW*demgroup	OMen 2Future	-8.03936953 B	3.41516478	-2.35	0.0189	-14.74797790	-1.33076117	**new Beta8**
sexMW*demgroup	OMen 3Current	0.00000000 B	.	.	.	.	.	
sexMW*demgroup	1Women 1None	0.00000000 B	.	.	.	.	.	
sexMW*demgroup	1Women 2Future	0.00000000 B	.	.	.	.	.	
sexMW*demgroup	1Women 3Current	0.00000000 B	.	.	.	.	.	

NOTE: The X'X matrix has been found to be singular, and a generalized inverse was used to solve the normal equations. Terms whose estimates are followed by the letter 'B' are not uniquely estimable.

Translation: "Hey, I noticed that you only needed two contrasts to distinguish your three demgroup categories, so I didn't put in the third one that would have been redundant. I also noticed that you only needed one contrast to distinguish your two sexMW categories, so I didn't put in the second one that

would have been redundant. So your reference group is the category that has the dots instead (which reversed the meaning of sexMW), k? Oh, and the "B" is there to remind you that these are each conditional given the interaction, k?"

So how DO we describe the interaction? First, let's look at the results that go with the <u>marginal main effects (means and mean differences) given above</u>:

```
Least Squares Means at age85=0, grip9=0
             cognition        Standard    H0:LSMEAN=0     H0:LSMean1=LSMean2
sexMW          LSMEAN           Error       Pr > |t|     t Value    Pr > |t|
OMen       23.0612690       0.9251161         <.0001        4.41      <.0001
1Women     17.6153984       0.7888452         <.0001
```

Marginal sex difference: averaged across the three dementia groups (so not helpful in understanding the interaction)

```
Least Squares Means at age85=0, grip9=0
              cognition        Standard                    LSMEAN
demgroup        LSMEAN           Error    Pr > |t|        Number
1None        27.6323493       0.4780479      <.0001             1
2Future      21.6585828       0.9283335      <.0001             2
3Current     11.7240690       1.4442532      <.0001             3
```

```
   Least Squares Means for Effect demgroup
   t for H0: LSMean(i)=LSMean(j) / Pr > |t|
i/j             1             2             3
   1                    5.744436      10.51128
                          <.0001        <.0001
   2      -5.74444                    5.797667
            <.0001                      <.0001
   3      -10.5113      -5.79767
            <.0001        <.0001
```

Marginal dementia difference: averaged across the two sex groups (so still not helpful in understanding the interaction). Note that only t-values and p-values are given (not the estimates of the difference or those SEs).

The tables below provide all possible pairwise cell differences:

```
Least Squares Means at age85=0, grip9=0
                       cognition        Standard                    LSMEAN
sexMW     demgroup       LSMEAN           Error    Pr > |t|         Number
OMen      1None       29.0701463       0.7484992      <.0001              1
OMen      2Future     23.0142449       1.4927601      <.0001              2
OMen      3Current    17.0994158       2.1402181      <.0001              3
1Women    1None       26.1945523       0.6388340      <.0001              4
1Women    2Future     20.3029208       1.1186333      <.0001              5
1Women    3Current     6.3487222       1.9478805      0.0012              6
```

This is also not helpful in understanding the interaction **because we need to know how these differences differ**! Note that only t-values and p-values are given (not the estimates of the difference or those SEs).

```
                  Least Squares Means for Effect sexMW*demgroup
                  t for H0: LSMean(i)=LSMean(j) / Pr > |t|
i/j             1             2             3             4             5             6
   1                    3.70363      5.332284       2.84364      6.483439      10.82948
                          0.0002        <.0001        0.0046        <.0001        <.0001
   2      -3.70363                    2.286576      -1.92636      1.446758      6.740621
            0.0002                      0.0226        0.0546        0.1485        <.0001
   3      -5.33228      -2.28658                    -4.05381      -1.32492      3.708001
            <.0001        0.0226                      <.0001        0.1858        0.0002
   4      -2.84364      1.926363      4.053807                    4.610903        9.7831
            0.0046        0.0546        <.0001                      <.0001        <.0001
   5      -6.48344      -1.44676      1.324919       -4.6109                    6.232566
            <.0001        0.1485        0.1858        <.0001                      <.0001
   6      -10.8295      -6.74062       -3.708       -9.7831      -6.23257
            <.0001        <.0001        0.0002        <.0001        <.0001
```

BOLD is what we know so far from this output (i.e., what ANOVA gives by default)…

Dementia Group Means	Men Mean	Women Mean	***MARGINAL MEAN***	Sex Difference
None Mean	29.07	26.20	***27.62***	−2.87 (*p*=.0046)
Future Mean	23.01	20.30	***21.66***	−2.71 (*p*=.1485)
Current Mean	17.10	6.35	***11.72***	−10.75 (*p*=.0002)
MARGINAL	***23.03***	***17.62***		***−5.45 (p<.0001)***
Dementia Group Differences	**Men Diff**	**Women Diff**	***MARGINAL DIFF***	**Simple Effect Difference**
None-Future Diff	−6.06 (*p*=.0002)	−5.90 (*p*<.0001)	***−5.97 (p<.0001)***	A =0.16 (*p*=.9368)
None-Current Diff	−11.97 (*p*<.0001)	−19.85 (*p*<.0001)	***−15.91 (p<.0001)***	B= −7.88 (*p*=.0095)
Future-Current Diff	−5.91 (*p*=.0226)	−13.95 (*p*<.0001)	***−9.93 (p<.0001)***	C= −8.04 (*p*=.0189)

So what we know from the default traditional ANOVA output (bolded in the table below) is that the effect of sex differs somehow by dementia group (but not whether there is an effect of sex within each dementia group) OR that the effect of dementia differs somehow by sex (but not whether there is an effect of dementia per sex). Super helpful, huh? This is why we need the rest of our code. To find out the differences within condition (simple effects), some ANOVA-trained-only people may turn to separate models (e.g., select only men, examine the effect of dementia group; repeat with women). However, that model will have a different residual variance, and thus the test may not be the same as it would be when done correctly in a full model. So let's see how to ask for all the relevant simple effects from the SAME model instead… **using SLICE within LSMEANS**!

sexMW*demgroup Effect Sliced by demgroup for cognition

demgroup	DF	Sum of Squares	Mean Square	F Value	Pr > F
1None	1	695.197526	695.197526	8.09	0.0046
2Future	1	179.949615	179.949615	2.09	0.1485
3Current	1	1182.058190	1182.058190	13.75	0.0002

Total Variation Accounted For

demgroup	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 95% Confidence Limits	
1None	0.0105	0.0092	0.0003	0.0336
2Future	0.0027	0.0014	0.0000	0.0181
3Current	0.0178	0.0165	0.0025	0.0455

These are the results of "SLICE" which give multivariate Wald tests for simple effects of each predictor at each level of the other predictor named on SLICE.

These results for the DF=2 dementia group difference for each sex were found in the previous versions of this model using CONTRAST in SAS or TEST in STATA.

sexMW*demgroup Effect Sliced by sexMW for cognition

sexMW	DF	Sum of Squares	Mean Square	F Value	Pr > F
0Men	2	3213.270817	1606.635408	18.69	<.0001
1Women	2	9140.125913	4570.062957	53.16	<.0001

Total Variation Accounted For

sexMW	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 95% Confidence Limits	
0Men	0.0485	0.0458	0.0179	0.0857
1Women	0.1379	0.1351	0.0872	0.1894

Parameter	Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits	
DF=1 simple effects of sex (already given by LSMEANS above and SLICES above)						
Sex Diff for None	-2.8755941	1.01123720	-2.84	0.0046	-4.8620266	-0.8891616
Sex Diff for Fut	-2.7113241	1.87406883	-1.45	0.1485	-6.3926673	0.9700192
Sex Diff for Cur	-10.7506936	2.89932314	-3.71	0.0002	-16.4460040	-5.0553832
DF=1 simple effects of dementia group (already given by LSMEANS above)						
None-Fut Diff for Men	-6.0559015	1.63512607	-3.70	0.0002	-9.2678755	-2.8439275
None-Fut Diff for Women	-5.8916315	1.27776082	-4.61	<.0001	-8.4016120	-3.3816510
None-Cur Diff for Men	-11.9707305	2.24495370	-5.33	<.0001	-16.3806247	-7.5608364
None-Cur Diff for Women	-19.8458301	2.02858306	-9.78	<.0001	-23.8306947	-15.8609655
Fut-Cur Diff for Men	-5.9148291	2.58676242	-2.29	0.0226	-10.9961581	-0.8335000
Fut-Cur Diff for Women	-13.9541986	2.23891711	-6.23	<.0001	-18.3522347	-9.5561625
Differences in simple effects = interactions not given above						
A: Sex Effect differ between None and Fut?	0.1642700	2.07047524	0.08	0.9368	-3.9028859	4.2314259
A: None-Fut Effect differ by Sex?	0.1642700	2.07047524	0.08	0.9368	-3.9028859	4.2314259
B: Sex Effect differ between None and Cur?	-7.8750995	3.02453647	-2.60	0.0095	-13.8163738	-1.9338253
B: None-Cur Effect differ by Sex?	-7.8750995	3.02453647	-2.60	0.0095	-13.8163738	-1.9338253
C: Sex Effect differ between Fut and Cur?	-8.0393695	3.41516478	-2.35	0.0189	-14.7479779	-1.3307612
C: Fut-Cur Effect differ by Sex?	-8.0393695	3.41516478	-2.35	0.0189	-14.7479779	-1.3307612

Moral of the story: Given a significant interaction, it is likely that the marginal tests will be misleading and/or will not help you test any meaningful hypotheses. Get simple effects differences!

Example Results Section Using SAS Output [notes about what also to include]:

We estimated a general linear model (as shown in Equation 1) to examine the extent to which cognition could be predicted from main effects of age (centered such that 0 = 85 years), grip strength (centered such that 0 = 9 pounds per square inch), sex (0 = men, 1= women), and dementia status (none vs. future; none vs. current), as well as an interaction between age and grip strength, and an interaction between sex and dementia status. The model accounted for a significant amount of variance in cognition, $F(8,541) = 28.77$, MSE = 85.97, $p < .0001$, $R^2 = .298$. Table 2 provides the model results, including the fixed effects estimated directly in the model, as well as their linear combinations in order to provide simple slopes by which to describe the sex by dementia group interaction. Two types of effect sizes are given in Table 2. First, semi-partial eta-squared (sr^2) estimates (i.e., the amount of unique variance accounted for) are given for the simple effects of age and grip strength, as well as the marginal age by grip strength interaction, the marginal effect of sex (averaging across dementia group), the marginal effect of dementia group (averaging across sex), and the marginal sex by dementia group interaction. Second, Cohen's *d* standardized mean differences computed from the *t*-values for each estimated fixed effect or linear combination thereof are given for the simple effects of sex, dementia group, and their interaction contrasts.

Equation 1:

$$\boldsymbol{Cognition_i = \beta_0 + \beta_1(Age_i - 85) + \beta_2(Grip_i - 9) + \beta_3(SexMW_i) + \beta_4(DemNF_i) + \beta_5(DemNC_i) + \beta_6(Age_i - 85)(Grip_i - 9) + \beta_7(SexMW_i)(DemNF_i) + \beta_8(SexMW_i)(DemNC_i) + e_i}$$

Results from this model can be interpreted as follows. The intercept $\beta_0 = 29.07$ is the expected cognition outcome for an 85-year-old man with 9 pounds of grip strength who will not be diagnosed with dementia later in the study.

[See example 6 or Hoffman (2015) ch. 2 for an example description of the age by grip interaction, which would go here]

The main and interactive effects of sex by dementia group are presented next, as illustrated in Figure X [reference first plot given above], in which the sex differences are shown by the vertical distances between the lines, and the dementia group differences are shown by the differences within the lines. First, with respect to the marginal main effects (assuming balanced sample sizes), men and women differed significantly as averaged across dementia groups, $F(1, 541) = 19.45$, MSE = 85.97, $p < .001$, the dementia groups differed significantly as averaged across sex, $F(2, 541) = 64.62$, MSE = 85.97, $p < .001$, and these marginal main effects were qualified by a significant sex by dementia group interaction, $F(2, 541) = 3.49$, MSE = 85.97, $p = .031$. To better understand the interaction, simple effects and their differences (i.e., interaction contrasts) are reported next.

First, there was a significant simple main effect of sex $\beta_3 = -2.88$ such that in the no dementia group, cognition was significantly lower by 2.88 in women than in men. The sex difference in cognition was equivalent in no dementia and

future dementia groups, as shown by the nonsignificant sex by no dementia vs. future dementia interaction $\beta_7 = 0.16$. However, the resulting sex difference in cognition favoring men in the future dementia group of $\beta_3 + \beta_7 = -2.88 + 0.16 = -2.71$ was not significant, likely a result of the small number of persons with future dementia (only 20% of the sample). In addition, the sex difference in cognition was significantly larger in the current dementia group than in the no dementia group, as shown by the significant sex by no dementia vs. current dementia interaction $\beta_8 = -7.88$, and the resulting sex difference in the current dementia group of $\beta_3 + \beta_8 = 2.88 - 7.88 = -10.75$ was also significant. The sex difference in cognition was also significantly larger in the current dementia group than in the future dementia group, as found by $\beta_8 - \beta_7 = -7.88 - 0.16 = -8.04$.

With respect to differences among the dementia groups, a significant omnibus group difference was found both in men, $F(2, 541) = 18.69$, MSE = 85.97, $p < .001$, and in women, $F(2, 541) = 53.16$, MSE = 85.97, $p < .001$. More specifically, cognition was significantly lower in the future dementia than no dementia group both in men, $\beta_4 = -6.06$, and in women, $\beta_4 + \beta_7 = -6.06 + 0.16 = -5.89$. This group difference was equivalent across sexes, as indicated by the nonsignificant sex by no dementia vs. future dementia interaction $\beta_4 = 0.16$. Cognition was also significantly lower in the current dementia than no dementia group both in men, $\beta_5 = -11.97$, and in women, $\beta_5 + \beta_8 = -11.97 - 7.88 = -19.85$. This group difference was significantly larger in women, as indicated by the sex by no dementia vs. current dementia interaction $\beta_8 = -7.88$. Finally, cognition was also significantly lower in the current dementia group than future diagnosis group both in men, $\beta_5 - \beta_4 = -11.97 + 6.06 = -5.91$, and in women, $\beta_5 + \beta_8 - \beta_4 - \beta_7 = -11.97 - 7.88 + 6.06 + 0.16 = -13.95$. This group difference was significantly larger in women, as indicated by the additional interaction contrast of $\beta_8 - \beta_7 = -7.88 - 0.16 = -8.04$.

Table 2: Model Results (bold values indicate $p < .001$)

	Model Effects	Est	SE	$p <$	η^2	Cohen's d
Model for the Means						
β_0	Intercept	29.07	0.75	.001		
β_1	Age Slope (0 = 85 years)	−0.33	0.12	.005	**.010**	
β_2	Grip Strength Slope (0 = 9 lbs)	0.62	0.15	.001	**.023**	
β_6	Age by Grip Interaction	0.12	0.04	.003	**.012**	
	Sex (0 = Men, 1 = Women) Differences:					
β_3	No Diagnosis	−2.88	1.01	.005		**−.121**
$\beta_3 + \beta_7$	Future Diagnosis	−2.71	1.87	.149		−.062
$\beta_3 + \beta_8$	Current Diagnosis	−10.75	2.90	.001		**−.157**
	Dementia Group Differences:					
	None vs. Future Diagnosis					
β_4	Men	−6.06	1.64	.001		**−.157**
$\beta_4 + \beta_7$	Women	−5.89	1.28	.001		**−.194**
β_7	Sex by None vs. Future	0.16	2.07	.937		.003
	None vs. Current Diagnosis					
β_5	Men	−11.97	2.25	.001		**−.223**
$\beta_5 + \beta_8$	Women	−19.85	2.02	.001		**−.388**
β_8	Sex by None vs. Current	−7.88	3.02	.010		**−.111**
	Future vs. Current Diagnosis					
$-\beta_4 + \beta_5$	Men	−5.91	2.59	.023		**−.098**
$-\beta_4 + \beta_5 - \beta_7 + \beta_8$	Women	−13.95	2.24	.001		**−.259**
$-\beta_7 + \beta_8$	Sex by Future vs. Current	−8.04	3.42	.019		**−.101**
	Sex Marginal Main Effect				**.025**	
	Dementia Marginal Main Effect				**.165**	
	Sex by Dementia Marginal Interaction				**.009**	

Note: η^2 = semipartial eta-squared for amount of model R^2 due to that predictor.